Random Non-homogeneous linear system examples

(from 4/25/2013)

```
> with(LinearAlgebra):
```

Let's solve the following system...

```
> diff(x(t),t) = <<2,0>|<1,2>>.x(t)+<<t,exp(2*t)>>;
```

$$\frac{d}{dt}\,x(t) = \begin{bmatrix} 2 & 1 \\ 0 & 2 \end{bmatrix} . x(t) + \begin{bmatrix} t \\ e^{2t} \end{bmatrix} \quad (1)$$

First, we need a fundamental matrix. Since this is a "constant coefficient" problem, we can actually find such a matrix. However, it is not diagonalizable (in fact our coefficient matrix is a 2 by 2 Jordan block). So we just need to compute the matrix exponential.

```
> Diag := <<2,0>|<0,2>>;
  Nilp := <<0,0>|<1,0>>;

  A := Diag + Nilp;
```

$$Diag := \begin{bmatrix} 2 & 0 \\ 0 & 2 \end{bmatrix}$$

$$Nilp := \begin{bmatrix} 0 & 1 \\ 0 & 0 \end{bmatrix}$$

$$A := \begin{bmatrix} 2 & 1 \\ 0 & 2 \end{bmatrix} \quad (2)$$

We can exponentiate the diagonal piece and use the definition of the matrix exponential to compute the nilpotent part. This gives us...

```
> X := <<exp(2*t),0>|<0,exp(2*t)>>.(IdentityMatrix(2)+Nilp*t);
```

$$X := \begin{bmatrix} e^{2t} & t\,e^{2t} \\ 0 & e^{2t} \end{bmatrix} \quad (3)$$

Or we could compute our fundamental matrix X directly...

```
> X := MatrixExponential(A*t);
```

$$X := \begin{bmatrix} e^{2t} & t\,e^{2t} \\ 0 & e^{2t} \end{bmatrix} \quad \textbf{(4)}$$

Now we need to use variation of parameters to find a particular solution. It is $X(t)\int X(t)^{-1} f(t)\,dt$. Let's find $X(t)^{-1} f(t)$.

```
> X^(-1).<<t,exp(2*t)>>;
```

$$\begin{bmatrix} \frac{t}{e^{2t}} - t \\ 1 \end{bmatrix} \quad \textbf{(5)}$$

We need to integrate this vector (term by term). The "proper" way to do this is with the "map" command. It "maps" the command "int" with argument "t" (the variable of integration) onto each entry of our matrix. In other words, it integrates the matrix/vector term by term.

```
> map(int, X^(-1).<<t,exp(2*t)>>, t);
```

$$\begin{bmatrix} -\frac{1}{2}\frac{t}{e^{2t}} - \frac{1}{4\,e^{2t}} - \frac{1}{2}t^2 \\ t \end{bmatrix} \quad \textbf{(6)}$$

Putting this altogether we have...

```
> x_p := simplify(X . map(int, X^(-1).<<t,exp(2*t)>>, t));
```

$$x\_p := \begin{bmatrix} -\frac{1}{2}t - \frac{1}{4} + \frac{1}{2}t^2\,e^{2t} \\ t\,e^{2t} \end{bmatrix} \quad \textbf{(7)}$$

The general solution is then...

```
> x = X.<<C_1,C_2>>+x_p;
```

$$x = \begin{bmatrix} e^{2t}\,C\_1 + t\,e^{2t}\,C\_2 - \frac{1}{2}t - \frac{1}{4} + \frac{1}{2}t^2\,e^{2t} \\ e^{2t}\,C\_2 + t\,e^{2t} \end{bmatrix} \quad \textbf{(8)}$$

Next, let's solve $y''' - 4\,y' = t$.

First, we'll do it the easy way (use undetermined coefficients).

Our characteristic equation is...

```
> r^3-4*r=0;
  solve(r^3-4*r=0);
```

$$r^3 - 4\,r = 0$$

$$0, 2, -2 \qquad (9)$$

So the homogeneous solution is...

```
> y_h := C_1*exp(0*t)+C_2*exp(2*t)+C_3*exp(-2*t);
```

$$y\_h := C\_1 + e^{2\,t}\,C\_2 + e^{-2\,t}\,C\_3 \qquad (10)$$

The right hand side of our equation (the non-homogeneous part) is "t". This is part of the homogeneous solution set $\{1, t\}$. But notice that we already have 1 as a homogeneous solution. Thus the set from which we're building our guess must be shifted by t: $\{t, t^2\}$. Thus our template (guess) solution is $y_p = A\,t^2 + B$. Let's define this and plug it into our differential equation.

```
> # undefine A (which was a matrix in the pervious example).
  A := 'A';
```

$$A := A \qquad (11)$$

```
> y_p := A*t^2+B*t;
```

$$y\_p := A\,t^2 + B\,t \qquad (12)$$

```
> diff(y_p,t$3)-4*diff(y_p,t)=t;
```

$$-8\,A\,t - 4\,B = t \qquad (13)$$

Equating coefficients we get...

```
> {-8*A=1,-4*B=0};
  solve({-8*A=1,-4*B=0});
```

$$\{-8\,A = 1, -4\,B = 0\}$$

$$\left\{A = -\frac{1}{8}, B = 0\right\} \qquad (14)$$

Now that we've finished the difficult task of solving those equations, we can write down our particular solution and then the general solution.

```
> A := -1/8;
  B := 0;

  'y_p' = y_p;
```

$$A := -\frac{1}{8}$$

$$B := 0$$

$$y\_p = -\frac{1}{8}\,t^2 \qquad (15)$$

```
> y = y_h + y_p;
```

$$y = C\_1 + e^{2t} C\_2 + e^{-2t} C\_3 - \frac{1}{8} t^2 \tag{16}$$

Now let's make life difficult and solve this equation again using variation of parameters. First, we solve the homogeneous equation (as above) and then write down our Wronskian matrices...

```
> y_1 := 1;
  y_2 := exp(2*t);
  y_3 := exp(-2*t);

  # The "full" Wronskian...
  W_0 := <<y_1, diff(y_1,t), diff(y_1,t,t)>|<y_2, diff(y_2,t), diff
  (y_2,t,t)>|<y_3, diff(y_3,t), diff(y_3,t,t)>>;

  # Leaving out y_1...
  W_1 := <<y_2, diff(y_2,t)>|<y_3, diff(y_3,t)>>;

  # Leaving out y_2...
  W_2 := <<y_1, diff(y_1,t)>|<y_3, diff(y_3,t)>>;

  # Leaving out y_3...
  W_3 := <<y_1, diff(y_1,t)>|<y_2, diff(y_2,t)>>;
```

$$y\_1 := 1$$

$$y\_2 := e^{2t}$$

$$y\_3 := e^{-2t}$$

$$W\_0 := \begin{bmatrix} 1 & e^{2t} & e^{-2t} \\ 0 & 2e^{2t} & -2e^{-2t} \\ 0 & 4e^{2t} & 4e^{-2t} \end{bmatrix}$$

$$W\_1 := \begin{bmatrix} e^{2t} & e^{-2t} \\ 2e^{2t} & -2e^{-2t} \end{bmatrix}$$

$$W\_2 := \begin{bmatrix} 1 & e^{-2t} \\ 0 & -2e^{-2t} \end{bmatrix}$$

$$W\_3 := \begin{bmatrix} 1 & e^{2t} \\ 0 & 2e^{2t} \end{bmatrix} \tag{17}$$

Keep in mind that $v_i = \int \frac{(-1)^{n+i} \det(W_i) g(t)}{\det(W_0)} dt$ where W_0 is the full Wronskian matrix, $g(t)$ is the non-homogeneous part of our equation, n is the order of the equation (in our case $n = 3$), and W_i is the Wronskian with the i^{th} homogeneous solution left out.

```
> v_1 := int( (-1)^(3+1)*Determinant(W_1)*t/Determinant(W_0), t);
  v_2 := int( (-1)^(3+2)*Determinant(W_2)*t/Determinant(W_0), t);
  v_3 := int( (-1)^(3+3)*Determinant(W_3)*t/Determinant(W_0), t);
```

$$v\_1 := -\frac{1}{8}t^2$$

$$v\_2 := -\frac{1}{32}\frac{1+2t}{e^{2t}}$$

$$v\_3 := \frac{1}{32}\frac{-1+2t}{e^{-2t}} \qquad (18)$$

```
> y_p := simplify(v_1*y_1 + v_2*y_2 + v_3*y_3);
```

$$y\_p := -\frac{1}{8}t^2 - \frac{1}{16} \qquad (19)$$

This is a different particular solution (it includes an extra "-1/16"). While not exactly optimal, this is ok. Notice that the 2 particular solutions are just off by a homogeneous solution (C_1=-1/16, $C_2 = C_3 = 0$).

Our general solution is then...

```
> y = y_h + y_p;
```

$$y = C\_1 + e^{2t}C\_2 + e^{-2t}C\_3 - \frac{1}{8}t^2 - \frac{1}{16} \qquad (20)$$

Now for no good reason, let's solve this equation one more time using a different method. Let's convert our 3rd order DE to a system of 3 first order DEs.

Let $x_1 = y$, $x_2 = x_1' = y'$, $x_3 = x_2' = y''$ and so $x_3' = y'''$. We then get $x_3' - 4\,x_2 = t$ (from our original equation). This then becomes the system...

```
> diff(x(t),t) = <<0,0,0>|<1,0,4>|<0,1,0>>.x(t)+<<0,0,t>>;
```

$$\frac{d}{dt}x(t) = \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 4 & 0 \end{bmatrix} . x(t) + \begin{bmatrix} 0 \\ 0 \\ t \end{bmatrix} \qquad (21)$$

Our (constant) coefficient matrix is...

```
> A := <<0,0,0>|<1,0,4>|<0,1,0>>;
```

$$A := \begin{bmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 4 & 0 \end{bmatrix} \qquad (22)$$

```
> Eigenvectors(A);
```

$$\begin{bmatrix} 0 \\ 2 \\ -2 \end{bmatrix}, \begin{bmatrix} 1 & \frac{1}{4} & \frac{1}{4} \\ 0 & \frac{1}{2} & -\frac{1}{2} \\ 0 & 1 & 1 \end{bmatrix} \quad \textbf{(23)}$$

Notice that the Eigenvectors matrix came back "full" (or that we have 3 distinct eigenvalues) so our coefficient matrix is diagonalizable. Thus we can find a fundamental matrix built from eigenvector, eigenvalue pairs. The eigenvector associated with eigenvalue 0 should be multiplied by $e^{0t} = 1$, the one associated with 2 should be multiplied by e^{2t} and the one associated with -2 should be multiplied by e^{-2t}. Thus we get the following fundamental matrix...

```
> X := <<1,0,0>|<exp(2*t),2*exp(2*t),4*exp(2*t)>|<exp(-2*t),-2*exp
  (-2*t),4*exp(-2*t)>>;
```

$$X := \begin{bmatrix} 1 & e^{2t} & e^{-2t} \\ 0 & 2e^{2t} & -2e^{-2t} \\ 0 & 4e^{2t} & 4e^{-2t} \end{bmatrix} \quad \textbf{(24)}$$

As before, we need a particular solution: $x_p = X\int X^{-1}f$

```
> f := <<0,0,t>>;
  x_p := simplify(X . map(int, X^(-1).f, t));
```

$$f := \begin{bmatrix} 0 \\ 0 \\ t \end{bmatrix}$$

$$x\_p := \begin{bmatrix} -\frac{1}{8}t^2 - \frac{1}{16} \\ -\frac{1}{4}t \\ -\frac{1}{4} \end{bmatrix} \quad \textbf{(25)}$$

Our general solution (for the system) is...

```
> x = X.<<C_1,C_2,C_3>> + x_p;
```

(26)

$$x = \begin{bmatrix} C\_1 + e^{2t} C\_2 + e^{-2t} C\_3 - \frac{1}{8} t^2 - \frac{1}{16} \\ 2 e^{2t} C\_2 - 2 e^{-2t} C\_3 - \frac{1}{4} t \\ 4 e^{2t} C\_2 + 4 e^{-2t} C\_3 - \frac{1}{4} \end{bmatrix} \quad \textbf{(26)}$$

The first entry is $x_1 = y$, the second entry is $x_2 = y'$, and the third entry is $x_3 = y''$.

It is interesting to notice that we got the same answer as when we used variation of parameters. This is no coincidence. In fact, our original variation of parameters technique always computes the first entry of associated the matrix solution. Why? Think about it. :)

One last thing...what if we want to solve $y''' - 4\,y' = g(t)$ (a general non-homogeneous equation)? Easy...

```
> f := <<0,0,g(t)>>;
  x_p := simplify(X . map(int, X^(-1).f, t));
  x = X.<<C_1,C_2,C_3>> + x_p;
```

$$f := \begin{bmatrix} 0 \\ 0 \\ g(t) \end{bmatrix}$$

$$x\_p := \begin{bmatrix} -\frac{1}{4} \int g(t)\,dt + \frac{1}{8} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) + \frac{1}{8} e^{-2t} \left(\int e^{2t} g(t)\,dt\right) \\ \frac{1}{4} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) - \frac{1}{4} e^{-2t} \left(\int e^{2t} g(t)\,dt\right) \\ \frac{1}{2} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) + \frac{1}{2} e^{-2t} \left(\int e^{2t} g(t)\,dt\right) \end{bmatrix}$$

$$x = \left[\left[C\_1 + e^{2t} C\_2 + e^{-2t} C\_3 - \frac{1}{4} \int g(t)\,dt + \frac{1}{8} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) + \frac{1}{8} e^{-2t} \left(\int e^{2t} g(t)\,dt\right)\right],\right.$$
$$\left[2 e^{2t} C\_2 - 2 e^{-2t} C\_3 + \frac{1}{4} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) - \frac{1}{4} e^{-2t} \left(\int e^{2t} g(t)\,dt\right)\right],$$
$$\left.\left[4 e^{2t} C\_2 + 4 e^{-2t} C\_3 + \frac{1}{2} e^{2t} \left(\int e^{-2t} g(t)\,dt\right) + \frac{1}{2} e^{-2t} \left(\int e^{2t} g(t)\,dt\right)\right]\right] \quad \textbf{(27)}$$

So for example: $y''' - 5\,y' = \ln(t)$ (cannot be solved using undetermined coefficients).

```
> f := <<0,0,ln(t)>>;
```

```
x_p := simplify(X . map(int, X^(-1).f, t));
x = X.<<C_1,C_2,C_3>> + x_p;
```

$$f := \begin{bmatrix} 0 \\ 0 \\ \ln(t) \end{bmatrix}$$

$$x\_p := \begin{bmatrix} -\frac{1}{4}\ln(t)\,t + \frac{1}{4}t - \frac{1}{16}e^{2t}\,\mathrm{Ei}(1, 2t) + \frac{1}{16}e^{-2t}\,\mathrm{Ei}(1, -2t) \\ -\frac{1}{4}\ln(t) - \frac{1}{8}e^{2t}\,\mathrm{Ei}(1, 2t) - \frac{1}{8}e^{-2t}\,\mathrm{Ei}(1, -2t) \\ -\frac{1}{4}e^{2t}\,\mathrm{Ei}(1, 2t) + \frac{1}{4}e^{-2t}\,\mathrm{Ei}(1, -2t) \end{bmatrix}$$

$$x = \left[\left[C\_1 + e^{2t}\,C\_2 + e^{-2t}\,C\_3 - \frac{1}{4}\ln(t)\,t + \frac{1}{4}t - \frac{1}{16}e^{2t}\,\mathrm{Ei}(1, 2t) + \frac{1}{16}e^{-2t}\,\mathrm{Ei}(1, -2t)\right], \left[2e^{2t}\,C\_2 - 2e^{-2t}\,C\_3 - \frac{1}{4}\ln(t) - \frac{1}{8}e^{2t}\,\mathrm{Ei}(1, 2t) - \frac{1}{8}e^{-2t}\,\mathrm{Ei}(1, -2t)\right], \left[4e^{2t}\,C\_2 + 4e^{-2t}\,C\_3 - \frac{1}{4}e^{2t}\,\mathrm{Ei}(1, 2t) + \frac{1}{4}e^{-2t}\,\mathrm{Ei}(1, -2t)\right]\right] \quad \textbf{(28)}$$

The general solution is

$$y = C\_1 + e^{2t}\,C\_2 + e^{-2t}\,C\_3 - \frac{1}{4}\ln(t)\,t + \frac{1}{4}t - \frac{1}{16}e^{2t}\,\mathrm{Ei}(1, 2t) + \frac{1}{16}e^{-2t}\,\mathrm{Ei}(1, -2t)$$

[The "Ei" functions are exponential integral functions. I guess it's good we didn't try to do this "by hand".
]